

4.8 Alternate Analysis as a Oneway ANOVA

- Suppose we have data from a two-factor factorial design. The following method can be used to perform a multiple comparison test to compare treatment means as well as Levene's Test to check the homogeneity of variance assumption.
- The main idea is to create a single factor having $a \times b$ levels and analyze the data as if you had a oneway ANOVA with $a \times b$ treatments.

Multiple Comparison Procedures

Suppose the researcher is interested in comparing the cell means from a two factor factorial design. The following method can be used to perform a multiple comparison procedure (MCP):

1. Create a single factor having $a \times b$ levels. For the 2×2 design example, create a single factor having four levels from the two levels of time ($T = 12, 18$) and the two levels of medium ($M = 1, 2$). In the SAS code, I called these levels 12_1, 12_2, 18_1, and 18_2.
2. Run Bonferroni's MCP (or any other MCP) on this single factor. For the 2×2 design, we reject all $H_0 : \mu_{ij} = \mu_{i'j'}$ except for $\mu_{12,2} = \mu_{12,1}$.

Levene's Test of the HOV Assumption


- In a twoway ANOVA, the HOV assumption implies that all $a \times b$ variances are equal. That is, we assume $\sigma_{11}^2 = \sigma_{12}^2 = \dots = \sigma_{ab}^2$ where σ_{ij}^2 is the variance associated with the errors for treatment combination (i, j) based on the two design factors.
- Suppose the researcher is interested in testing the HOV assumption that all of the $a \times b$ variances are equal in a two factor factorial design.
- The following method can be used to perform Levene's HOV Test:
 1. Create a single factor having $a \times b$ treatment levels. For the 2×2 design example, create a single factor having four levels from the two levels of time (12,18) and the two levels of medium (1,2). In the SAS code, I called these levels 12_1, 12_2, 18_1, and 18_2.
 2. Run Levene's HOV Test on this single factor. For the 2×2 design example, we fail to reject the HOV assumption (p -value = .1793).

Dependent Variable: growth

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	3	691.4583333	230.4861111	45.12	<.0001
Error	20	102.1666667	5.1083333		
Corrected Total	23	793.6250000			

R-Square	Coeff Var	Root MSE	growth Mean
0.871266	7.629240	2.260162	29.62500

Source	DF	Type III SS	Mean Square	F Value	Pr > F
time_med	3	691.4583333	230.4861111	45.12	<.0001



Levene's Test for Homogeneity of growth Variance ANOVA of Absolute Deviations from Group Means					
Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
time_med	3	6.3472	2.1157	1.80	0.1793
Error	20	23.4815	1.1741		

Alpha	0.05
Error Degrees of Freedom	20
Error Mean Square	5.108333
Critical Value of t	2.92712
Minimum Significant Difference	3.8196

Means with the same letter are not significantly different.			
Bon Grouping	Mean	N	time_med
A	37.167	6	18_1
B	32.000	6	18_2
C	26.000	6	12_2
C			
C	23.333	6	12_1

```

DM 'LOG; CLEAR; OUT; CLEAR;';

ODS GRAPHICS ON;
ODS PRINTER PDF file='C:\COURSES\ST541\TWOALT.PDF';
OPTIONS NODATE NONUMBER;

*****;
*** ALTERNATIVE ANALYSIS FOR A TOWAY ANOVA PLUS ***;
*** TESTING THE HOMOGENEITY OF VARIANCE ASSUMPTION ***;
*** AND MULTIPLE COMPARISON PROCEDURES ***;
*****;

DATA in;
  DO time_med = '12_1' , '12_2' , '18_1' , '18_2';
  DO rep = 1 to 6;
    INPUT growth @@; OUTPUT;
  END; END;
CARDS;
21 23 20 22 28 26      25 24 29 26 25 27
37 38 35 39 38 36      31 29 30 34 33 35
;
PROC GLM DATA=in;
  CLASS time_med;
  MODEL growth = time_med / SS3;
  MEANS time_med / BON HOVTEST=LEVENE(TYPE=ABS);
TITLE 'ALTERNATE ANOVA AND HOV TEST';
RUN;

```

4.9 Other Multiple Comparison Procedures

- You can also perform a MCP using the LSMEANS statement in Proc GLM in SAS. E.g., to perform a Bonferroni MCP:
 1. Include a **LSMEANS A*B** statement with option / **ADJUST=BON** for factors A and B.
 2. Reject $H_0 : \mu_{ij} - \mu_{i'j'}$ if the adjusted p -value in the matrix of p -values is $\leq \alpha$. This is equivalent to taking the p -value $\leq \alpha^*$ where $\alpha^* = \alpha/C$ and C is the number of comparisons made. In essence, we are just multiplying the individual test p -values by the number of comparisons, and then checking if the adjusted p -value is $< \alpha$.
- In SAS, it is possible to perform a test of the equality of cell means (i) across the levels of factor A for a specified level j of factor B and (ii) across the levels of factor B for a specified level i of factor A. These are called **tests of effects slices**.
- Tests of effects slices can be performed using the LSMEANS statement in Proc GLM in SAS.
 1. Include a statement of the form **LSMEANS A*B / SLICE = A ADJUST = BON** for a MCP of cell means across the levels of factor B for each level of factor A. The hypotheses tested for level i of factor A are

$H_0 :$	$H_1 :$
---------	---------
 2. Include a statement of the form **LSMEANS A*B / SLICE = B ADJUST = BON** for a MCP of cell means across the levels of factor A for each level of factor B. The hypotheses tested for level b of factor B are

$H_0 :$	$H_1 :$
---------	---------
- For a different MCP, just change BON to TUKEY, SIDAK, or SCHEFFE. If you do not include the ADJUST option, you will get Fisher's LSD test by default.

4.10 ANOVA for a 2 × 3 Factorial Design Example

- An experiment was run to investigate how the type of glass and the type of phosphorescent coating affects the brightness of a light bulb. The response variable is the current (in microamps) to obtain a specified brightness. The data are

		Phosphor Type		
		A	B	C
Glass Type	1	278	297	273
		291	304	284
		285	296	288
2	2	229	259	228
		235	249	225
		241	241	235

- Look at the difference in glass means across the levels of phosphor:

$$\bar{y}_{2,A} - \bar{y}_{1,A} = -49.7 \quad \bar{y}_{2,B} - \bar{y}_{1,B} = -49.3 \quad \bar{y}_{2,C} - \bar{y}_{1,C} = -52.3$$

The variability in these three differences ($MS_{glass*phosphor} = 4.0\bar{5}$) is small relative to the $MS_E = 44.\bar{2}$, so we fail to reject the null hypothesis that the interaction effects are equal.

- The Glass*Phosphor interaction is not significant (p -value = .9130). This is also obvious from the strong parallelism in the interaction plots.
- The Bonferroni MCT results are summarized below:

Glass/Phosphor	2 C	2 A	2 B	1 C	1 A	1 B
Mean	229.3	235.0	249.7	281.7	284.7	299.0

SAS Code and Output

```
DM 'LOG; CLEAR; OUT; CLEAR;';
ODS GRAPHICS ON;
ODS PRINTER PDF file='C:\COURSES\ST541\TOWAY2.PDF';
OPTIONS NODATE NONUMBER;

*****;
*** 2-FACTOR FACTORIAL (2x3) DESIGN ***;
*****;
DATA in; INPUT glass phosphor $ light @@; CARDS;
1 A 278 1 A 291 1 A 285 1 B 297 1 B 304 1 B 296
1 C 273 1 C 284 1 C 288 2 A 229 2 A 235 2 A 241
2 B 259 2 B 249 2 B 241 2 C 225 2 C 228 2 C 235
;
PROC GLM DATA=in ;
CLASS glass phosphor;
MODEL light = glass|phosphor / SS3 SOLUTION;
MEANS glass|phosphor / BON;
LSMEANS glass*phosphor / SLICE=glass SLICE=phosphor ADJUST=BON;;

ESTIMATE 'mu' intercept 1;

ESTIMATE 'glass=1' glass 1 -1 / divisor = 2 ;
ESTIMATE 'glass=2' glass -1 1 / divisor = 2 ;

ESTIMATE 'phosphor=A' phosphor 2 -1 -1 / divisor = 3 ;
ESTIMATE 'phosphor=B' phosphor -1 2 -1 / divisor = 3 ;
ESTIMATE 'phosphor=C' phosphor -1 -1 2 / divisor = 3 ;
```

```

ESTIMATE 'glass=1 phos=A' glass*phosphor 2 -1 -1 -2 1 1 / divisor = 6 ;
ESTIMATE 'glass=1 phos=B' glass*phosphor -1 2 -1 1 -2 1 / divisor = 6 ;
ESTIMATE 'glass=1 phos=C' glass*phosphor -1 -1 2 1 1 -2 / divisor = 6 ;
ESTIMATE 'glass=2 phos=A' glass*phosphor -2 1 1 2 -1 -1 / divisor = 6 ;
ESTIMATE 'glass=2 phos=B' glass*phosphor 1 -2 1 -1 2 -1 / divisor = 6 ;
ESTIMATE 'glass=2 phos=C' glass*phosphor 1 1 -2 -1 -1 2 / divisor = 6 ;

```

```

TITLE '(2 x 3) TWO FACTOR ANALYSIS OF VARIANCE';
RUN;

```

(2 x 3) TWO FACTOR ANALYSIS OF VARIANCE

The GLM Procedure

Variable: light




Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	5	12626.44444	2525.28889	57.10	<.0001
Error	12	530.66667	44.22222		
Corrected Total	17	13157.11111			

R-Square	Coeff Var	Root MSE	light Mean
0.959667	2.526375	6.649979	263.2222

Source	DF	Type III SS	Mean Square	F Value	Pr > F
glass	1	11450.88889	11450.88889	258.94	<.0001
phosphor	2	1167.44444	583.72222	13.20	0.0009
glass*phosphor	2	8.11111	4.05556	0.09	0.9130

Parameter	Estimate		Standard Error	t Value	Pr > t
Intercept	229.3333333	B	3.83936723	59.73	<.0001
glass 1	52.3333333	B	5.42968521	9.64	<.0001
glass 2	0.0000000	B	.	.	.
phosphor A	5.6666667	B	5.42968521	1.04	0.3172
phosphor B	20.3333333	B	5.42968521	3.74	0.0028
phosphor C	0.0000000	B	.	.	.
glass*phosphor 1 A	-2.6666667	B	7.67873446	-0.35	0.7344
glass*phosphor 1 B	-3.0000000	B	7.67873446	-0.39	0.7029
glass*phosphor 1 C	0.0000000	B	.	.	.
glass*phosphor 2 A	0.0000000	B	.	.	.
glass*phosphor 2 B	0.0000000	B	.	.	.
glass*phosphor 2 C	0.0000000	B	.	.	.

Parameter	Estimate	Standard Error	t Value	Pr > t
mu	263.222222	1.56741511	167.93	<.0001
glass=1	25.222222	1.56741511	16.09	<.0001
glass=2	-25.222222	1.56741511	-16.09	<.0001
phosphor=A	-3.388889	2.21665970	-1.53	0.1522
phosphor=B	11.111111	2.21665970	5.01	0.0003
phosphor=C	-7.722222	2.21665970	-3.48	0.0045
glass=1 phos=A	-0.388889	2.21665970	-0.18	0.8637
glass=1 phos=B	-0.555556	2.21665970	-0.25	0.8063
glass=1 phos=C	0.944444	2.21665970	0.43	0.6776
glass=2 phos=A	0.388889	2.21665970	0.18	0.8637
glass=2 phos=B	0.555556	2.21665970	0.25	0.8063
glass=2 phos=C	-0.944444	2.21665970	-0.43	0.6776




Alpha	0.05
Error Degrees of Freedom	12
Error Mean Square	44.22222
Critical Value of t	2.17881
Minimum Significant Difference	6.8302

Alpha	0.05
Error Degrees of Freedom	12
Error Mean Square	44.22222
Critical Value of t	2.77947
Minimum Significant Difference	10.671

Means with the same letter are not significantly different.			
Bon Grouping	Mean	N	glass
A	288.444	9	1
B	238.000	9	2

Means with the same letter are not significantly different.			
Bon Grouping	Mean	N	phosphor
A	274.333	6	B
B	259.833	6	A
B	255.500	6	C



Level of glass	Level of phosphor	N	light	
			Mean	Std Dev
1	A	3	284.666667	6.50640710
1	B	3	299.000000	4.35889894
1	C	3	281.666667	7.76745347
2	A	3	235.000000	6.00000000
2	B	3	249.666667	9.01849951
2	C	3	229.333333	5.13160144

The GLM Procedure
Least Squares Means
Adjustment for Multiple Comparisons: Bonferroni

glass	phosphor	light LSMEAN	LSMEAN Number
1	A	284.666667	1
1	B	299.000000	2
1	C	281.666667	3
2	A	235.000000	4
2	B	249.666667	5
2	C	229.333333	6

Least Squares Means for effect glass*phosphor Pr > t for H0: LSMean(i)=LSMean(j)						
Dependent Variable: light						
i/j	1	2	3	4	5	6
1		0.3237	1.0000	<.0001	0.0005	<.0001
2	0.3237		0.1161	<.0001	<.0001	<.0001
3	1.0000	0.1161		<.0001	0.0011	<.0001
4	<.0001	<.0001	<.0001		0.2890	1.0000
5	0.0005	<.0001	0.0011	0.2890		0.0420
6	<.0001	<.0001	<.0001	1.0000	0.0420	

The GLM Procedure
Least Squares Means

glass*phosphor Effect Sliced by glass for light					
glass	DF	Sum of Squares	Mean Square	F Value	Pr > F
1	2	514.888889	257.444444	5.82	0.0171
2	2	660.666667	330.333333	7.47	0.0078

The GLM Procedure
Least Squares Means

glass*phosphor Effect Sliced by phosphor for light					
phosphor	DF	Sum of Squares	Mean Square	F Value	Pr > F
A	1	3700.166667	3700.166667	83.67	<.0001
B	1	3650.666667	3650.666667	82.55	<.0001
C	1	4108.166667	4108.166667	92.90	<.0001

4.11 Example: Sample size determination and power estimation

Determine N given a nominal power level (Case 1) and determine power given N (Case 2) for a specified pattern of means or effects

- Suppose there are 6 treatments resulting from a 2×3 factorial design having n replicates, and based on a prior study we have estimates of the treatment means: $\mu_{11} = 35.1$, $\mu_{12} = 33.7$, $\mu_{13} = 30.2$, $\mu_{21} = 23.0$, $\mu_{22} = 25.9$, $\mu_{23} = 30.4$.
- Our prior estimate of σ is 1.4, and the significance level is set to $\alpha = .05$ for tests.
- For **Case 1**, determine the total sample size $N = 6n$ setting the power for the ANOVA F -tests for main effects and the interaction at levels $1 - \beta = .50, .80, .90,$ and $.95$. Also, determine N for several contrasts.
- For **Case 2**, determine the power $1 - \beta$ for the ANOVA F -tests for the main effects and the interaction when the total sample size $N = 18, 24, 30,$ and 36 . Also, determine power for the tests of several contrasts.

SAS code for Case 1: Determine N for a nominal power level

```
data twoway;
  input levelA $ levelB $ meanest;
datalines;
A1 B1 35.1  A1 B2 33.7  A1 B3 30.2  A2 B1 23.0  A2 B2 25.9  A2 B3 30.4
;
proc glmpower data=twoway;
  class levelA levelB;
  model meanest = levelA|levelB;
  contrast 'A1-A2' levelA 1 -1 ;
  contrast 'B1-B2' levelB 1 -1 0;
  contrast 'B1-B3' levelB 1 0 -1;
  contrast 'B2-B3' levelB 0 1 -1;
  contrast 'A11-A12' levelB 1 -1 0 levelA*levelB 1 -1 0 0 0 0;
  contrast 'A12-A23' levelA 1 -1 levelB 0 1 -1 levelA*levelB 0 1 0 0 0 -1;
  power
    stddev = 1.4
    alpha = 0.05
    ntotal = .
    power = .5 .8 .9 .95 ;
title 'Determining design size for given power and mean estimates';
title2 'for a twoway (2 x 3) ANOVA';

proc glmpower data=twoway;
  class levelA levelB;
  model meanest = levelA|levelB;
  contrast 'A1-A2' levelA 1 -1 ;
  contrast 'B1-B2' levelB 1 -1 0;
  contrast 'B1-B3' levelB 1 0 -1;
  contrast 'B2-B3' levelB 0 1 -1;
  contrast 'A11-A12' levelB 1 -1 0 levelA*levelB 1 -1 0 0 0 0;
  contrast 'A12-A23' levelA 1 -1 levelB 0 1 -1 levelA*levelB 0 1 0 0 0 -1;
  power
    stddev = 1.4
    alpha = 0.05
    ntotal = 18 24 30 36
    power = . ;
title 'Determining power for a given design size and mean estimates';
title2 'for a twoway (2 x 3) ANOVA';
run;
```

SAS output for Case 1: Determine N for a nominal power level

The GLMPOWER Procedure

Fixed Scenario Elements

Dependent Variable meanest
 Alpha 0.05
 Error Standard Deviation 1.4

Computed N Total

Index	Type	Source	Nominal Power	Test DF	Error DF	Actual Power	N Total
1	Effect	levelA	0.50	1	6	>.999	12
2	Effect	levelA	0.80	1	6	>.999	12
3	Effect	levelA	0.90	1	6	>.999	12
4	Effect	levelA	0.95	1	6	>.999	12
5	Effect	levelB	0.50	2	36	0.520	42
6	Effect	levelB	0.80	2	72	0.818	78
7	Effect	levelB	0.90	2	96	0.915	102
8	Effect	levelB	0.95	2	114	0.954	120
9	Effect	levelA*levelB	0.50	2	6	0.992	12
10	Effect	levelA*levelB	0.80	2	6	0.992	12
11	Effect	levelA*levelB	0.90	2	6	0.992	12
12	Effect	levelA*levelB	0.95	2	6	0.992	12
13	Contrast	A1-A2	0.50	1	6	>.999	12
14	Contrast	A1-A2	0.80	1	6	>.999	12
15	Contrast	A1-A2	0.90	1	6	>.999	12
16	Contrast	A1-A2	0.95	1	6	>.999	12
17	Contrast	B1-B2	0.50	1	78	0.508	84
18	Contrast	B1-B2	0.80	1	162	0.805	168
19	Contrast	B1-B2	0.90	1	216	0.900	222
20	Contrast	B1-B2	0.95	1	270	0.952	276
21	Contrast	B1-B3	0.50	1	30	0.562	36
22	Contrast	B1-B3	0.80	1	60	0.830	66
23	Contrast	B1-B3	0.90	1	78	0.910	84
24	Contrast	B1-B3	0.95	1	96	0.954	102
25	Contrast	B2-B3	0.50	1	180	0.507	186
26	Contrast	B2-B3	0.80	1	366	0.801	372
27	Contrast	B2-B3	0.90	1	492	0.901	498
28	Contrast	B2-B3	0.95	1	612	0.951	618
29	Contrast	A11-A12	0.50	1	48	0.547	54
30	Contrast	A11-A12	0.80	1	96	0.823	102
31	Contrast	A11-A12	0.90	1	126	0.908	132
32	Contrast	A11-A12	0.95	1	156	0.955	162
33	Contrast	A12-A23	0.50	1	6	0.507	12
34	Contrast	A12-A23	0.80	1	18	0.883	24
35	Contrast	A12-A23	0.90	1	24	0.947	30
36	Contrast	A12-A23	0.95	1	30	0.977	36

SAS output for Case 2: Determine power for a given N

The GLMPower Procedure

Fixed Scenario Elements

Dependent Variable	meanest
Alpha	0.05
Error Standard Deviation	1.4

Computed Power

Index	Type	Source	N Total	Test DF	Error DF	Power
1	Effect	levelA	18	1	12	>.999
2	Effect	levelA	24	1	18	>.999
3	Effect	levelA	30	1	24	>.999
4	Effect	levelA	36	1	30	>.999
5	Effect	levelB	18	2	12	0.215
6	Effect	levelB	24	2	18	0.296
7	Effect	levelB	30	2	24	0.375
8	Effect	levelB	36	2	30	0.450
9	Effect	levelA*levelB	18	2	12	>.999
10	Effect	levelA*levelB	24	2	18	>.999
11	Effect	levelA*levelB	30	2	24	>.999
12	Effect	levelA*levelB	36	2	30	>.999
13	Contrast	A1-A2	18	1	12	>.999
14	Contrast	A1-A2	24	1	18	>.999
15	Contrast	A1-A2	30	1	24	>.999
16	Contrast	A1-A2	36	1	30	>.999
17	Contrast	B1-B2	18	1	12	0.137
18	Contrast	B1-B2	24	1	18	0.174
19	Contrast	B1-B2	30	1	24	0.210
20	Contrast	B1-B2	36	1	30	0.246
21	Contrast	B1-B3	18	1	12	0.296
22	Contrast	B1-B3	24	1	18	0.394
23	Contrast	B1-B3	30	1	24	0.483
24	Contrast	B1-B3	36	1	30	0.562
25	Contrast	B2-B3	18	1	12	0.088
26	Contrast	B2-B3	24	1	18	0.104
27	Contrast	B2-B3	30	1	24	0.120
28	Contrast	B2-B3	36	1	30	0.135
29	Contrast	A11-A12	18	1	12	0.204
30	Contrast	A11-A12	24	1	18	0.268
31	Contrast	A11-A12	30	1	24	0.330
32	Contrast	A11-A12	36	1	30	0.389
33	Contrast	A12-A23	18	1	12	0.755
34	Contrast	A12-A23	24	1	18	0.883
35	Contrast	A12-A23	30	1	24	0.947
36	Contrast	A12-A23	36	1	30	0.977